F08GEF (SSPTRD/DSPTRD) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F08GEF (SSPTRD/DSPTRD) reduces a real symmetric matrix to tridiagonal form, using packed storage.

2 Specification

```
SUBROUTINE F08GEF(UPLO, N, AP, D, E, TAU, INFO)
ENTRY      ssptrd(UPLO, N, AP, D, E, TAU, INFO)
INTEGER            N, INFO
real               AP(*), D(*), E(*), TAU(*)
CHARACTER*1        UPLO
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine reduces a real symmetric matrix A, held in packed storage, to symmetric tridiagonal form T by an orthogonal similarity transformation: $A = QTQ^T$.

The matrix Q is not formed explicitly but is represented as a product of $n-1$ elementary reflectors (see the Chapter Introduction for details). Routines are provided to work with Q in this representation (see Section 8).

4 References

[1] Golub G H and van Loan C F (1996) *Matrix Computations* Johns Hopkins University Press (3rd Edition), Baltimore

5 Parameters

1: UPLO — CHARACTER*1 *Input*

On entry: indicates whether the upper or lower triangular part of A is stored as follows:

if UPLO = 'U', then the upper triangular part of A is stored;
if UPLO = 'L', then the lower triangular part of A is stored.

Constraint: UPLO = 'U' or 'L'.

2: N — INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: AP(*) — ***real*** array *Input/Output*

Note: the dimension of the array AP must be at least max(1,N*(N+1)/2).

On entry: the n by n symmetric matrix A, packed by columns. More precisely, if UPLO = 'U', the upper triangle of A must be stored with element a_{ij} in AP$(i+j(j-1)/2)$ for $i \leq j$; if UPLO = 'L', the lower triangle of A must be stored with element a_{ij} in AP$(i+(2n-j)(j-1)/2)$ for $i \geq j$.

On exit: A is overwritten by the tridiagonal matrix T and details of the orthogonal matrix Q.

4: D(∗) — ***real*** array *Output*

Note: the dimension of the array D must be at least max(1,N).

On exit: the diagonal elements of the tridiagonal matrix T.

5: E(∗) — ***real*** array *Output*

Note: the dimension of the array E must be at least max(1,N−1).

On exit: the off-diagonal elements of the tridiagonal matrix T.

6: TAU(∗) — ***real*** array *Output*

Note: the dimension of the array TAU must be at least max(1,N−1).

On exit: further details of the orthogonal matrix Q.

7: INFO — INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed tridiagonal matrix T is exactly similar to a nearby matrix $A + E$, where

$$\| E \|_2 \leq c(n)\epsilon \| A \|_2,$$

$c(n)$ is a modestly increasing function of n, and ϵ is the ***machine precision***.

The elements of T themselves may be sensitive to small perturbations in A or to rounding errors in the computation, but this does not affect the stability of the eigenvalues and eigenvectors.

8 Further Comments

The total number of floating-point operations is approximately $\frac{4}{3}n^3$.

To form the orthogonal matrix Q this routine may be followed by a call to F08GFF (SOPGTR/DOPGTR):

```
      CALL SOPGTR (UPLO,N,AP,TAU,Q,LDQ,WORK,INFO)
```

To apply Q to an n by p real matrix C this routine may be followed by a call to F08GGF (SOPMTR/DOPMTR). For example,

```
      CALL SOPMTR ('Left',UPLO,'No Transpose',N,P,AP,TAU,C,LDC,WORK,
     +             INFO)
```

forms the matrix product QC.

The complex analogue of this routine is F08GSF (CHPTRD/ZHPTRD).

9 Example

To reduce the matrix A to tridiagonal form, where

$$A = \begin{pmatrix} 2.07 & 3.87 & 4.20 & -1.15 \\ 3.87 & -0.21 & 1.87 & 0.63 \\ 4.20 & 1.87 & 1.15 & 2.06 \\ -1.15 & 0.63 & 2.06 & -1.81 \end{pmatrix},$$

using packed storage.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08GEF Example Program Text
*     Mark 16 Release. NAG Copyright 1992.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=8)
*     .. Local Scalars ..
      INTEGER          I, INFO, J, N
      CHARACTER        UPLO
*     .. Local Arrays ..
      real             AP(NMAX*(NMAX+1)/2), D(NMAX), E(NMAX-1),
     +                 TAU(NMAX-1)
*     .. External Subroutines ..
      EXTERNAL         ssptrd
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08GEF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read A from data file
*
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
         ELSE IF (UPLO.EQ.'L') THEN
            READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
         END IF
*
*        Reduce A to tridiagonal form
*
         CALL ssptrd(UPLO,N,AP,D,E,TAU,INFO)
*
*        Print tridiagonal form
*
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Diagonal'
         WRITE (NOUT,99999) (D(I),I=1,N)
         WRITE (NOUT,*) 'Off-diagonal'
         WRITE (NOUT,99999) (E(I),I=1,N-1)
      END IF
```

```
      STOP
*
99999 FORMAT (1X,8F9.4)
      END
```

9.2 Program Data

```
F08GEF Example Program Data
  4                           :Value of N
  'L'                         :Value of UPLO
  2.07
  3.87  -0.21
  4.20   1.87   1.15
 -1.15   0.63   2.06  -1.81   :End of matrix A
```

9.3 Program Results

```
 F08GEF Example Program Results

 Diagonal
    2.0700   1.4741  -0.6492  -1.6949
 Off-diagonal
   -5.8258   2.6240   0.9163
```